Spectral estimates on the sphere: interpolation inequalities
J. Dolbeault , M.J. Esteban, A. Laptev

A priori estimates: upper and lower bounds

$$\mathbf{Integrate}\left[\left(1-x^2\right)^{\frac{d}{2}-1},\ \{x,-1,1\},\ \mathbf{Assumptions}\to d>0\right]$$

$$\mathbf{Integrate}\left[x^2\left(1-x^2\right)^{\frac{d}{2}-1},\ \{x,-1,1\},\ \mathbf{Assumptions}\to d>0\right]$$

$$\mathbf{Integrate}\left[(1+t\,x)^q\left(1-x^2\right)^{\frac{d}{2}-1},\ \{x,-1,1\},\ \mathbf{Assumptions}\to t>0\ \&\&\ t<1\ \&\&\ d>0\right]$$

$$\frac{\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{d}{2}\right]}{\mathrm{Gamma}\left[\frac{1+d}{2}\right]}$$

$$\frac{\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{d}{2}\right]}{2\ \mathrm{Gamma}\left[\frac{3+d}{2}\right]}$$

$$\sqrt{\pi}\ \mathrm{Gamma}\left[\frac{d}{2}\right]\ \mathrm{Hypergeometric2F1Regularized}\left[\frac{1-q}{2},\ -\frac{q}{2},\ \frac{1+d}{2},\ t^2\right]$$

$$\mathbf{Z[d\_]} := \frac{\sqrt{\pi}\ \mathbf{Gamma}\left[\frac{d}{2}\right]}{\mathbf{Gamma}\left[\frac{1+d}{2}\right]}$$

$$\mathbf{Z2[d\_]} := \frac{\sqrt{\pi}\ \mathbf{Gamma}\left[\frac{d}{2}\right]}{2\ \mathbf{Gamma}\left[\frac{3+d}{2}\right]}$$

$$\mathbf{g[d\_,\ q\_,\ t\_]} := \sqrt{\pi}\ \mathbf{Gamma}\left[\frac{d}{2}\right]\ \mathbf{Hypergeometric2F1Regularized}\left[\frac{1-q}{2},\ -\frac{q}{2},\ \frac{1+d}{2},\ t^2\right]$$

$$\mathbf{f[d\_,\ q\_,\ \alpha\_,\ t\_]} := \frac{\mathbf{Z[d]}^{\frac{2}{q}-1}}{\alpha}\ \frac{\left(\alpha+t^2\right)\mathbf{Z[d]}+t^2\,(\alpha-1)\ \mathbf{Z2[d]}}{\mathbf{g[d,\ q,\ t]}^{\frac{2}{q}}}$$

```
h[d_, q_, α_] := FindMinimum[f[d, q, α, t], {t, 0, 1}]
```

```
Plot[f[3, 3, 3.2, t], {t, 0, 1}]
h[3, 3, 3.2]
```




```
{0.999043, {t → 0.359211}}
```

```
{Plot[{1, h[3, 3, a][[1]]}, {a, 0, 10}],
 Plot[{1, h[3, 3, a][[1]]}, {a, 2.9, 3.2}], Plot[{1, h[3, 3, a][[1]]}, {a, 3, 3.02}]}
```




```
Upper[d_, q_, α_] := α h[d, q, α][[1]]
```

$$\text{Lower}[d\_, q\_, \alpha\_] := \text{If}\left[\alpha < \frac{d}{q-2}, \alpha, \left(\frac{d}{q-2}\right)^{\vartheta} \alpha^{1-\vartheta} \text{ /. } \vartheta \to d\,\frac{q-2}{2\,q}\right]$$

```
P1 = Plot[{a, Upper[3, 3, a], Lower[3, 3, a]}, {a, 0, 15},
  PlotStyle → {{Black, Thickness[0.005], Dashed}, {Black, Dashing[Tiny],
     Thickness[0.005]}, {Black, Dashing[Tiny], Thickness[0.005]}}]
```



Asymptotic regime

```
S[d_] := 2 π^(d/2) / Gamma[d/2];

Fn[a_, q_, d_, rmax_, η_, PR_] := Module[
  {M = Evaluate[u[s] /. NDSolve[{v'[r] + (d - 1) v[r]/r + Abs[u[r]]^(q-2) u[r] - u[r] == 0,
       u'[r] == v[r], v[η] == (a - Abs[a^(q-2)] a)/d η, u[η] == a + (a - Abs[a^(q-2)] a)/d η^2/2},
      {u, v}, {r, η, rmax}]]}, Plot[M, {s, η, rmax}, PlotRange → PR]]

ufinal[a_, q_, d_, rmax_, η_] := u[rmax] /.
  NDSolve[{v'[r] + (d - 1) v[r]/r + Abs[u[r]]^(q-2) u[r] - u[r] == 0, u'[r] == v[r], v[η] ==
     (a - Abs[a^(q-2)] a)/d η, u[η] == a + (a - Abs[a^(q-2)] a)/d η^2/2}, {u, v}, {r, η, rmax}][[1]]

H[a_, q_, d_, rmax_, η_] := Log[1 + u[rmax]^2 + v[rmax]^2] /.
  NDSolve[{v'[r] + (d - 1) v[r]/r + Abs[u[r]]^(q-2) u[r] - u[r] == 0, u'[r] == v[r], v[η] ==
     (a - Abs[a^(q-2)] a)/d η, u[η] == a + (a - Abs[a^(q-2)] a)/d η^2/2}, {u, v}, {r, η, rmax}][[1]]
Nrm[q_, d_, a_, rmax_, η_] := {z[rmax], w2[rmax], w[rmax]} /.
  NDSolve[{v'[r] + (d - 1) v[r]/r + Abs[u[r]]^(q-2) u[r] - u[r] == 0,
     u'[r] == v[r], w'[r] == r^(d-1) Abs[u[r]]^q, w2'[r] == r^(d-1) Abs[u[r]]^2,
     z'[r] == r^(d-1) Abs[v[r]]^2, z[η] == ((a - Abs[a^(q-2)] a)/d)^2 η^(d+2)/(d + 2),
     w[η] == η^d/d a^q, w2[η] == η^d/d a^2, v[η] == (a - Abs[a^(q-2)] a)/d η,
     u[η] == a + (a - Abs[a^(q-2)] a)/d η^2/2}, {u, v, w, w2, z}, {r, η, rmax}][[1]]
QuotGN[q_, d_, a_, rmax_, η_] := Module[{M = Nrm[q, d, a, rmax, η]},
  (M[[1]] + M[[2]]) / M[[3]]^(2/q)]
```

```
Iter[a_, h_, q_, d_, rmax_, η_, b_, η1_, j_, Nmax_] :=
 Module[{M = {H[a + h, q, d, rmax, η], ufinal[a + h, q, d, rmax, η]}},
  If[Or[b < η1, j > Nmax], {j, N[a], b, M[[1]], N[h], QuotGN[q, d, a, rmax, η]},
   If[M[[2]] > 0, Iter[a + h, Abs[h], q, d, rmax, η, M[[1]], η1, j + 1, Nmax],
    Iter[a + h, -Abs[h / 2], q, d, rmax, η, M[[1]], η1, j + 1, Nmax]]]]

Init[a_,  h_, q_, d_, rmax_, η_, η1_, Nmax_] :=
 Iter[a, h, q, d, rmax, η, H[a, q, d, rmax, η], η1, 1, Nmax]

Conclusion[a_, h_, q_, d_, rmax_, η_, η1_, amin_, amax_, Nmax_] :=
 Module[{M = Init[a, h, q, d, rmax, η, η1, Nmax]},
  {M, Plot[H[aa, q, d, rmax, η], {aa, amin, amax}],
   Show[Fn[M[[2]], q, d, rmax, η, All]] }]

Conclusion[1, 1, 3, 3, 10, 10^-10, 10^-7, 1, 4, 200]
KGN = QuotGN[3, 3, 4.191680908203125`, 10, 10^-10]
```

$\{\{24,\ 4.19168,\ 4.52601\times10^{-8},\ 7.27902\times10^{-6},\ 0.0000305176,\ 2.75217\},$



,



$\}$

2.75217

```
Hasym[α_, d_, q_] := (S[d] / S[d + 1])^(1 - 2/q) KGN α^(1-ϑ) /. ϑ → d (q - 2)/(2 q)

P2 = Plot[Hasym[α, 3, 3], {α, 0, 30},
  PlotStyle → {Black, Dashing[{0.03, 0.02}], Thickness[0.005]}]
```



Computation of the initial datum for the Euler-Lagrange equations in dimension d=3 with q=3

```
ϵ = 10^-7;
d = 3;
q = 3;
ϑ = d (q - 2)/(2 q);
αmin = 0.1;
Off[NDSolve::ndinnt]
Off[ReplaceAll::reps]

δa = 0.1;

amin[α_] := If[α < 18, If[α < 7, Exp[-2.398361662890645`],
    Exp[-0.45122236251312486` α + 0.8601948747012291` -  δa]],
   Exp[-1.4216296940516822` - 0.28772080504742636` α - 2 δa]]

amax[α_] := If[α < 18, If[α < 7, Exp[2.8963427523371754` - 0.5992434878896886` α],
    Exp[-0.45122236251312486` α + 0.8601948747012291` + 10 δa]],
   Exp[-1.4216296940516822` - 0.28772080504742636` α + δa]]

Plot[{Log[amin[α]], Log[amax[α]]}, {α, 3, 30}]
```




```
F[α_] := Plot[
  u[π - ϵ] /. NDSolve[{u''[θ] + (d - 1) Cot[θ] u'[θ] + Abs[u[θ]]^(q-2) u[θ] - α u[θ] == 0,
     u'[ϵ] == (α a - Abs[a^(q-2)] a)/d ϵ, u[ϵ] == a + (α a - Abs[a^(q-2)] a)/d ϵ^2/2},
    {u, u'}, {θ, ϵ, π - ϵ}], {a, 0, 5}]

G[α_] := If[α < d/(q - 2), α^(1/(q-2)), a /. FindRoot[
    u[π - ϵ] /. NDSolve[{u''[θ] + (d - 1) Cot[θ] u'[θ] + Abs[u[θ]]^(q-2) u[θ] - α u[θ] == 0,
       u'[ϵ] == (α a - Abs[a^(q-2)] a)/d ϵ, u[ϵ] == a + (α a - Abs[a^(q-2)] a)/d ϵ^2/2},
      {u, u'}, {θ, ϵ, π - ϵ}], {a, amin[α], amax[α]}]]
```

```
Off[FindRoot::cvmit]
Off[FindRoot::brmp]
Off[FindRoot::lstol]
Plot[{Log[amin[α]], Log[amax[α]], Log[G[α]]}, {α, 3, 30}]
```




```
F[3.2]
G[3.2]
```




```
1.6259
```

```
Plot[G[α], {α, 0, 30}, PlotRange → All]
```



Computation of the optimal constant in dimension d=3 with q=3

```
H[α_] := If[α < d/(q - 2), {α}, Module[{a = G[α]}, (v[π - ϵ]/Z[d])^((q-2)/q) /.
    NDSolve[{u''[θ] + (d - 1) Cot[θ] u'[θ] + Abs[u[θ]]^(q-2) u[θ] - α u[θ] == 0,
      v'[θ] == Sin[θ]^(d-1) Abs[u[θ]]^q, u'[ϵ] == (α a - Abs[a^(q-2)] a)/d ϵ, u[ϵ] == a +
        (α a - Abs[a^(q-2)] a)/d ϵ^2/2, v[ϵ] == Abs[a^q]/d ϵ^d}, {u, u', v}, {θ, ϵ, π - ϵ}]]][[1]]
```

```
P3 = Plot[H[α], {α, 0, 30}, PlotStyle → {Black, Thick}]
```



Summary

```
Legende =
  {ListLinePlot[{{1, 9}, {3, 9}}, PlotStyle → {Black, Thickness[0.005], Dashed}],
   ListLinePlot[{{1, 10.5}, {3, 10.5}}, PlotStyle →
     {Black, Dashing[Tiny], Thickness[0.005]}], ListLinePlot[{{1, 12}, {3, 12}},
    PlotStyle → {Black, Dashing[{0.03, 0.02}], Thickness[0.005]}],
   ListLinePlot[{{1, 13.5}, {3, 13.5}}, PlotStyle → {Black, Thick}]};
```

```
Show[P1, P2, P3, Legende]
```



```
Show[P1, P2, P3, PlotRange → {{3, 10}, {3, 8}}]
```



$$\texttt{Plot}\left[\left\{1, \frac{\texttt{H}[\alpha]}{\texttt{Hasym}[\alpha, 3, 3]}\right\}, \{\alpha, 0, 50\}, \texttt{PlotRange} \to \texttt{All}, \texttt{PlotStyle} \to \{\texttt{Black}, \{\texttt{Black}, \texttt{Thick}\}\}\right]$$